

PS Ripple Study during the 2019 Fermilab Booster Experiment

*F. Schmidt*¹,

*J.S. Eldred, C. Jensen, J.L. Larson, H. Pfeffer, A.K. Triplett*²

Abstract

S.02 has been one of the experiment of the 2019 Fermilab Booster Studies. Its goal had been to evaluate the tune ripple on beam stability. A prerequisite is to measure the voltage ripple in both AC and DC mode. With a good knowledge of the Booster transmission line model one can determine how voltage ripple leads to tune modulation depth at the same ripple frequency.

Geneva, Switzerland
September 24, 2019

¹CERN, CH 1211 Geneva 23, Switzerland

²Fermilab, P.O. Box500, Batavia 60510, IL

1 Introduction

This report first discusses in the first chapter “Fermilab Booster Electric Model” with the single element, the cell definition and the model of all the Booster machine consisting of 48 cells.

In the next chapter the July transmission line modeling is presented between 1 Hz and 10 kHz. In August the measurement of the total Fermilab Booster have been done between 20 Hz and 2020 Hz both in open and closed mode.


In July during the 2019 Fermilab Booster Experiment the voltage ripple has been measured both in AC & DC mode.

In Appendix A the SPICE control file is presented.

In Appendix B a short overview of the SPS transmission line is given followed by the SPICE control file that has been adapted from the Fermilab Booster to the SPS case.

2 Fermilab Booster Electric Model


Simple Spice Model of Booster Components
 Based on Measurements of a single magnet done by Si Fang circa 1990



Si Fang D Magnet Model Values
 R1=12.8 mOhm
 RL1= 3kOhm, RL2=0.3 Ohm
 L1=9.7 mH, L2= 1.0 mH
 C1=C2=29 nF

Si Fang F Magnet Model Values
 R1=15.8 mOhm
 RL1=3 kOhm, RL2=0.3 Ohm
 L1=9.7 mH, L2=1.0 mH
 C1=C2=22 nF

Booster Choke
 (Path for offset current through Capacitor)



Si Fang Choke Model Values
 R1=26 mOhm, RP= 5k Ohm, RS=295 mOhm
 LP=LS=42 mH, k=0.998->Lair=84 uH
 CP=5 nF, C1=C2=2 nF, CA=CB= 92 nF

Figure 1: Fermilab Electric Models: D magnet, F magnet and Choke.

Fig. 1 shows the electric equivalent models D magnet, F magnet and the choke.




Figure 2: Fermilab Booster cell model.

Fig. 2 shows how the 3 electric models are hooked up to build one cell of the Fermilab Booster. Please note that there is separately a huge capacity bank of $8341 \mu\text{F}$ that is the crucial part of the resonant circuit of 15 Hz between the magnet and choke part. In Ref. [1] it has been shown how the capacity has to be set to achieve two resonance conditions at the same time.




Figure 3: Full Fermilab Booster Model of 48 cells.

In Fig. 3 the connections of all 48 cells of the Fermilab Booster is depicted. It also shows the location of the four power stations. Of particular importance is the fact that each of the 48 chokes are connected with all other 47 chokes and at each end. To this end there is a secondary winding in each

choke. This set-up has been implemented to minimize the unbalancing of each individual choke.

3 2019 Booster Transmission Line Modeling and Measuring

Using the SPICE control file (see Appendix A) the impedance of the full Fermilab Booster both for the closed and open circuits are depicted in the upper part of Fig. 4. The smallest resonance frequency is seen at ~ 9 Hz, the second at ~ 15 Hz (closed circuit), the third at ~ 85 Hz (both circuits) and so on. Please note also that at ~ 3 kHz both curves coincide.




Figure 4: Transmission Line for Open and Closed Booster Circuit (Range 1 Hz - 10 kHz): Upper Plot - Impedance of Machine, Lower Plot - Impedance of D & F Magnets.

Instead, the lower part of Fig. 4 shows the impedance with respect to the D & F magnets added up over all 48 cells. In other words, one considers only the current that actually flows through any of the D or F magnets, not the current that flows out of the power supply. Only the current flowing through the D and F magnets contributes to magnetic field in the aperture. Current flowing through the choke or capacitor do not directly contribute to field in the aperture but are a load on the power supply.

Together with the measured voltage ripple one can use this impedance to calculate the tune modulation depth [3] that could substantially limit beam stability.

There are two further sources of losses that are not contributing to magnetic field: a damping resistor, presently in a stage of study (see Fig. 5), or that part of the current that is lost in the vacuum chamber (the latter is quite relevant for the SPS but most likely not very significant for the Fermilab Booster). On the other hand, we do find some more damping in the measurements (see below) and that might be due to the fact that structures with eddy current losses like vacuum chambers have not been considered.



Figure 5: Potential damping resistor (50Ω , $250W$) for each cell of Fermilab Booster.

Including such a damping resistors of 50 Ohms in every cell leads to a damping over all the resonance regime of the transmission lines (see Fig. 4) above some 100 Hz. To this end please compare with the model results *without damping resistor* of Fig. 4. It remains to be seen if an installation of the damping resistors is justified in view of a performance upgrade of the Fermilab Booster.




Figure 6: Transmission Line with damping resistors (RD=50 Ohm) for Open and Closed Booster Circuit (Range 1 Hz - 10 kHz): Upper Plot - Impedance of Machine, Lower Plot - Impedance of D & F Magnets.





Figure 7: Comparison of Booster Transmission Line Measurements: Upper graph Closed circuit; lower graph Open circuit respectively.

Fig. 7 shows the comparison of the Booster impedance measurements (August 2019) with the model expectations: upper part for the closed and lower part for the open circuits respectively. The measurements have been done in 2 shifts: from 20 Hz till 2020 Hz and 5 Hz till 50 Hz. The overlap gives

an indication of the measurement reproducibility. The overall agreement is quite satisfying however with the following exceptions:

- At ~ 240 Hz and ~ 650 Hz one finds outliers.
- The impedance of the Booster was measured with a small signal impedance analyzer, approximately 0.5 Vrms. Looking at the terminals without the analyzer present there was 20 mVrms, 80 mV peak to peak of 120 Hz and 180 Hz signal present. This background signal will cause issues with the measured impedance near these frequencies. Apparently for that reason, the expected peak at some 220 Hz (open circuit) is not found in the experiment.
- The two outliers in the experiments (310 Hz/240 Hz and 610 Hz/650 Hz in closed and open mode respectively) are most likely due auxiliary equipment like pumps that are never switched off even when everything else is de-energized.
- Apparently, the capacitance to ground needs to be adjusted for a better modeling of the open circuit.
- In the measurements, one finds stronger damping. Extending the measurements of the D, F and Choke individual magnets to higher frequencies may help resolve this stronger damping.


4 Voltage Ripple Measurements

The voltage ripple measurement has been done in two ways: a direct measurement at the power supply station and using a probe. The latter method should ensure that external noise is diminished substantially.

Fig. 8 shows the expected and very large resonance line at 15 Hz due to AC machine configuration. In both methods (Fig. 8 part a) and part b), respectively) the ripple lines are “washed” out in three regimes: at lower frequencies up to about 400 Hz; around 720 Hz and 1440 Hz, the latter two expected from the 12 pole power supply convertors. For the AC mode the probe measurements shows the same noisy behaviour as does the direct measurement mode.

In the DC machine configuration one finds with the direct measurement unexpected lines at 540 Hz and 1260 Hz while the expected frequency lines at 720 Hz and 1440 Hz are hardly measurably (see part a) of Fig. 9). This has been the motivation to try out the probe measurement. Indeed, part b) of Fig. 9 shows that the fake lines at 540 Hz and 1260 Hz are reduced by a factor of 10, while the expected lines at 720 Hz and 1440 Hz are increased by a factor of 10. The probe measurement is most likely the trustworthy measurement.

a)



b)





Figure 8: Voltage Ripple Measurement in AC Mode: Part a) Original Measurement directly at PS Station; Part b) Measurement with Probe.

a)



b)




Figure 9: Voltage Ripple Measurement in DC Mode: Part a) Original Measurement directly at PS Station; Part b) Measurement with Probe.

Bibliography

- [1] J. Ryk, "GRADIENT MAGNET POWER SUPPLY FOR THE FERMILAB 8-GeV PROTON SYNCHROTRON", FERMILAB-Pub-74/85 0323.000, (Submitted to IEEE Transactions on Industrial Applications), August 1974.

- [2] P. Burla, D. Cornuet, K. Fischer, P. Leclere, F. Schmidt, "Power supply ripple study at the SPS", CERN SL/94-11 (AP) (1994) (Revised January 1996), AIP Conference Proceedings 326 (1995), <http://cern.ch/Frank.Schmidt/report/power1.pdf>.
- [3] M. Furman and F. Schmidt, "REDUCTION OF THE DYNAMICAL APERTURE DUE TO TUNE MODULATION", CERN SPS/89-1 (AMS), SSCL-A-6.

Appendix

A SPICE Booster Model

In this appendix one finds the latest SPICE control file used for the analysis in this report and in particular the data to produce. figs. 4 and 7.

```

48-CELL BOOSTER RING MODEL, AC Stimulation at PS1; PS2, 3 & 4 shorted out
* 7/1/19
* All models included
*
* Doing Transient Simulation need time waveformss
* Power Supply as a subcircuit. Filter in subcircuit
* XPS1 3 4 101 PS
* XPS2 1 2 102 PS
* XPS3 7 8 103 PS
* XPS4 5 6 104 PS
*
* PS Voltages using sine waves
* Sin( Voff Vamp Freq Tdelay Tdamping)
* VPS1 101 0 Sin( 520.0 322.0 15.0) DC 0
* VPS2 102 0 Sin( 520.0 322.0 15.0) DC 0
* VPS3 103 0 Sin( 520.0 322.0 15.0) DC 0
* VPS4 104 0 Sin( 520.0 322.0 15.0) DC 0
*
* Doing small signal frequency analysis need AC source
*
RCOM 4 0 1000k
* RCOM 4 0 1m
* Make RCOM=1000k for open, 1m for short
VMEA 100 3 DC 0
VTST 100 0 DC 0 AC 1
RPS2 1 2 1m
RPS3 7 8 1m
RPS4 5 6 1m
*
* All choke secondaries are connected to nodes T3 and T4
*
RX1 T3 0 100000k
RX2 T4 0 100000k
*
* The current through VBEAM represents the sum of all the currents
* through the D and F magnets, as opposed to the total current
* from the PS which is VMEA
*
RM MM 0 100000k
VBEAM MM 0 DC 0
*
*SECTION A CELLS
*
*
X1 A1 A2 0 T3 T4 MM CELLO3
* X1 2 A2 0 T3 T4 MM CELLO3
X2 A2 A3 0 T3 T4 MM CELLO3

```

```

X3   A3 A4   0 T3 T4 MM CELLO3
X4   A4 A5   0 T3 T4 MM CELLO3
*
X5   A5 A6   0 T3 T4 MM CELLO3
X6   A6 A7   0 T3 T4 MM CELLO3
X7   A7 A8   0 T3 T4 MM CELLO3
X8   A8 A9   0 T3 T4 MM CELLO3
*
X9   A9 A10  0 T3 T4 MM CELLO3
X10  A10 A11 0 T3 T4 MM CELLO3
X11  A11 A12 0 T3 T4 MM CELLO3
* X12 A12 3   0 T3 T4 MM CELLO3
X12  A12 B1  0 T3 T4 MM 3 4 CELLO3PS
*
*
*SECTION B CELLS
*
* X13 4 B2   0 T3 T4 MM CELLO3
X13  B1 B2   0 T3 T4 MM CELLO3
X14  B2 B3   0 T3 T4 MM CELLO3
X15  B3 RM2  0 T3 T4 MM CELLO3
** Reference Magnet connection
VC   RM1 RM2 DC 0
XRM  B4 RM1 0 REFMAG
X16  B4 B5   0 T3 T4 MM CELLO3
*
X17  B5 B6   0 T3 T4 MM CELLO3
X18  B6 B7   0 T3 T4 MM CELLO3
X19  B7 B8   0 T3 T4 MM CELLO3
X20  B8 B9   0 T3 T4 MM CELLO3
*
X21  B9 B10  0 T3 T4 MM CELLO3
X22  B10 B11 0 T3 T4 MM CELLO3
X23  B11 B12 0 T3 T4 MM CELLO3
* X24 B12 5   0 T3 T4 MM CELLO3
X24  B12 C1  0 T3 T4 MM 5 6 CELLO3PS
*
*SECTION C CELLS
*
*
* X25 6 C2   0 T3 T4 MM CELLO3
X25  C1 C2   0 T3 T4 MM CELLO3
X26  C2 C3   0 T3 T4 MM CELLO3
X27  C3 C4   0 T3 T4 MM CELLO3
X28  C4 C5   0 T3 T4 MM CELLO3
*
X29  C5 C6   0 T3 T4 MM CELLO3
X30  C6 C7   0 T3 T4 MM CELLO3
X31  C7 C8   0 T3 T4 MM CELLO3
X32  C8 C9   0 T3 T4 MM CELLO3
*
X33  C9 C10  0 T3 T4 MM CELLO3
X34  C10 C11 0 T3 T4 MM CELLO3
X35  C11 C12 0 T3 T4 MM CELLO3
* X36 C12 7   0 T3 T4 MM CELLO3
X36  C12 D1  0 T3 T4 MM 7 8 CELLO3PS
*
*SECTION D CELLS
*
*
```

```

* X37 8 D2 0 T3 T4 MM CELLO3
X37 D1 D2 0 T3 T4 MM CELLO3
X38 D2 D3 0 T3 T4 MM CELLO3
X39 D3 D4 0 T3 T4 MM CELLO3
X40 D4 D5 0 T3 T4 MM CELLO3
*
X41 D5 D6 0 T3 T4 MM CELLO3
X42 D6 D7 0 T3 T4 MM CELLO3
X43 D7 D8 0 T3 T4 MM CELLO3
X44 D8 D9 0 T3 T4 MM CELLO3
*
X45 D9 D10 0 T3 T4 MM CELLO3
X46 D10 D11 0 T3 T4 MM CELLO3
X47 D11 D12 0 T3 T4 MM CELLO3
* X48 D12 1 0 T3 T4 MM CELLO3
X48 D12 A1 0 T3 T4 MM 1 2 CELLO3PS
*
*
* SUBCIRCUITS SUBCIRCUITS
*
* Version of cell with current monitor for D and F magnet
* These currents effect beam, while I(C1) and I(CHOKE) are not
* seen by beam
.SUBCKT CELLO3 T1 T2 T3 T4 T5 T6
* Bus between cells is not a good guess, just a placeholder
R1 T1 D 10m
L1 D C 10u
CC C T3 4n
VDFM C E DC 0
*
X1 E A T3 DMAGSF
X2 A B T3 FMAGSF
X3 B T2 T4 T5 T3 CHOKE
* What resonant capacitor value to choose?
* Si Fang model use 8341
* C1 T2 B 8341u
* George Krafczyk model use 7590
C1 T2 B 7590u
* Dan Wolff and Howie Pfeffer Model use ?
* C1 T2 B 7892u
RD T1 T2 50000k
* Ripple in D and F magnet
FMEAS 0 T6 VDFM .0208
RMEAS T6 0 1000000k
.ENDS CELLO3
*
.SUBCKT CELLO3PS T1 T2 T3 T4 T5 T6 P1 P2

* Bus between cells is not a good guess, just a placeholder
R1 T1 D 10m
L1 D C 10u
CC C T3 4n
X1 C G T3 DMAGSF
LP1 G H 0.5m
VDM H P1 DC 0
LP2 P2 J 0.5m
VFM J K DC 0
X2 K B T3 FMAGSF
X3 B T2 T4 T5 T3 CHOKE
* Si Fang model use 8341

```

```

C1      T2 B 8341u
RD      T1 T2 50000k
*
FMEAS1 0 T6 VDM .0104
FMEAS2 0 T6 VFM .0104
RMEAS  T6 0 1000000k
.ENDS CELLO3PS
*
* Si Fang Writeup Values
.SUBCKT DMAGSF 1 2 3
R1      1 4 12.8m
L1      4 5 9.7m
RL1     4 2 3k
L2      5 2 1.0m
RL2     5 2 0.3
C1      1 3 29n
C2      2 3 29n
.ENDS DMAGSF
*
* Dan W & Howie P D Magnet Model, different freq response
.SUBCKT DMAGDW 1 2 3
R1      1 4 17.3m
L1      4 5 6.1m
RL1     4 5 38
L2      5 2 4.1m
RL2     5 2 3k
C1      1 3 29n
C2      2 3 29n
.ENDS DMAGDW
*
*** George Krafczyk D Magnet Model, done for power loss
.SUBCKT DMAGGK 1 2 3
R1      1 4 17.3m
L1      4 5 6.7m
L2      5 2 4.5m
RL2     5 2 45
C1      1 3 29n
C2      2 3 29n
.ENDS DMAGGK

* Si Fang Writeup Values
*
.SUBCKT FMAGSF 1 2 3
R1      1 4 15.8m
L1      4 5 9.7m
RL2     4 2 3k
L2      5 2 1.0m
RL1     5 2 0.3
C1      1 3 22n
C2      2 3 22n
.ENDS FMAGSF
*
* Dan W & Howie P F Magnet Model, different freq response
.SUBCKT FMAGDW 1 2 3
R1      1 4 17.3m
L1      4 5 6.1m
RL2     4 5 38
L2      5 2 4.1m
RL1     5 2 3k
C1      1 3 22n

```

```

C2      2 3 22n
.ENDS FMAGDW
*
* George K F Magnet Model, done for power loss
.SUBCKT FMAGGK 1 2 3
R1      1 4 17.3m
L1      4 5 6.7m
L2      5 2 4.5m
RL1     4 2 45
C1      1 3 22n
C2      2 3 22n
.ENDS FMAGGK
*
.SUBCKT CHOKE 1 2 3 4 5
R1      1 6 26m
LP      6 2 42m
RP      6 2 5k
CP      6 2 5n
CA      6 7 92n
CB      2 4 92n
C1      1 5 2n
C2      2 5 2n
R2      3 7 295m
LS      7 4 42m
KPS     LP LS 0.998
.ENDS CHOKE
*
.SUBCKT CHOKENC 1 2 3 4 5
R1      1 6 26m
LP      6 2 42m
RP      6 2 5k
CP      6 2 5n
CA      6 7 92n
CB      2 4 92n
C1      1 5 2n
C2      2 5 2n
R2      3 7 295m
.ENDS CHOKENC
*
* Georges Power loss model, no coupling
.SUBCKT CHGK 1 2 3 4 5
R1      1 6 36m
L1      6 8 27.4m
L2      8 2 16.5m
RL1     8 2 105
CP      6 2 5n
CA      6 7 92n
CB      2 4 92n
C1      1 5 2n
C2      2 5 2n
R2      3 7 295m
.ENDS CHGK

.SUBCKT REFMAG 1 2 3
R1      1 4 9.64m
L2      4 2 2.32m
R2      4 2 1800
CR2     4 2 3n
C1      1 3 2.4n
C2      2 3 2.4n

```

```

.ENDS REFMAG
*
.SUBCKT PS    P    N    C
E1    P1 N1 C 0 1
L1    P1 P 0.9m
L2    N1 N 0.9m
R1    P 1 1.76
C1    1 N 2422u
C2    P N 486u
RG1   P 0 50k
RG2   N 0 50k
CP    P 0 1u
CN    N 0 1u
rrrc  C 0 1000.0
.ENDS PS

.control
* AC analysis
* DEC (LIN)  Np  Fstart  Fstop
ac  dec  200 1 10k
* Transient analysis
*   TSTEP TSTOP <TSTART <TMAX>>
* tran 10u 10 8 200u
set length=5000
set width=132
set nopage
* plot v(2) v(A4) v(A8) v(A12)
plot mag(v(3)/i(vmea) ) mag(v(3)/i(vbeam) )
plot ph(v(3)/i(vmea) ) ph(v(3)/i(vbeam) )
* print mag(v(3)/i(vmea) ) ph(v(3)/i(vmea) ) mag(v(3)/i(vbeam) ) ph(v(3)/i(vbeam) ) > boosterzo.txt
print mag(v(3)/i(vmea) ) ph(v(3)/i(vmea) ) mag(v(3)/i(vbeam) ) ph(v(3)/i(vbeam) ) > boostshort.txt
* print mag(v(3)/i(vmea) ) > boosterzmago.txt
* print ph(v(3)/i(vmea) ) > boosterzpho.txt
.endc

```

B SPICE SPS Model

Starting with the SPICE model of the Fermilab Booster it was straight forward to downgrade it to the case of the SPS quadrupole chain.




Figure B.1: Equivalent Electric Model of a single Quadrupole of the SPS (taken from Ref. [2]).

Fig. B.1 shows the model of a single SPS quadrupole cell. Apparently it has been sufficient to set just 5 parameters:

- Residual resistor in series with the inductance $RL=14.2\text{m}\Omega$
- Inductance $L=14.7\text{mH}$

- Damping resistor $R_d=24\Omega$
- $2 \times$ resistor to ground $R_C=M\Omega$
- $2 \times$ capacitance to ground $C=nF$

The damping resistor has been fitted to the measurement of the full chain of quadrupoles and in fact it has been lowered to include the effect of the vacuum chamber. No attempt has been made to consider a partial shielding of the inductance due the vacuum chamber.




Figure B.2: Comparison of Measurement and Modeling of the full chain of SPS quadrupoles (taken from Ref. [2]).

Fig. B.2 shows excellent agreement between measurements and modeling, except for some deviation at 300 Hz for the open circuit measurements. Also there are some kinks at ~ 2000 & ~ 3000 that maybe nonphysical.

Taken the SPICE model of the Fermilab Booster (see section Appendix A), downgrading it to this simple SPS model and appending 108 identical cells we would expect agreement with transmission line modeling according to Ref. [2]. Next we present the SPICE model used for the SPS.

108-CELL SPS Quad Model

* 8/11/19

* Credit goes to Chris Jensen to provide the SPICE

* Model for the Fermilab Booster

* Modified to fit to the SPS case F. Schmidt

*

* Doing small signal frequency analysis need AC source

*

* RCOM 4 0 0.001m

RCOM 4 0 1000k

* Make RCOM=1000k for open, 1m for short

VMEA 100 3 DC 0

VTST 100 0 DC 0 AC 1

* RPS2 1 2 1m

* RPS3 7 8 1m

* RPS4 5 6 1m

RPS2 D73 A1 1m

RPS3 C13 D1 1m

RPS4 B13 C1 1m

*

* The current through VBEAM represents the sum of all the currents

* through the D and F magnets, as opposed to the total current

* from the PS which is VMEA

*

RM MM 0 100000k

VBEAM MM 0 DC 0

*

*SECTION A CELLS

*

*

X1 A1 A2 0 T3 T4 MM CELLSPS

X2 A2 A3 0 T3 T4 MM CELLSPS

X3 A3 A4 0 T3 T4 MM CELLSPS

X4 A4 A5 0 T3 T4 MM CELLSPS

*

X5 A5 A6 0 T3 T4 MM CELLSPS

X6 A6 A7 0 T3 T4 MM CELLSPS

X7 A7 A8 0 T3 T4 MM CELLSPS

X8 A8 A9 0 T3 T4 MM CELLSPS

*

X9 A9 A10 0 T3 T4 MM CELLSPS

X10 A10 A11 0 T3 T4 MM CELLSPS

X11 A11 A12 0 T3 T4 MM CELLSPS

X12 A12 3 0 T3 T4 MM CELLSPS

*

*

*SECTION B CELLS

*

X13 4 B2 0 T3 T4 MM CELLSPS

X14 B2 B3 0 T3 T4 MM CELLSPS

X15 B3 B4 0 T3 T4 MM CELLSPS

X16 B4 B5 0 T3 T4 MM CELLSPS

*

X17 B5 B6 0 T3 T4 MM CELLSPS

X18 B6 B7 0 T3 T4 MM CELLSPS

X19 B7 B8 0 T3 T4 MM CELLSPS

X20 B8 B9 0 T3 T4 MM CELLSPS

*

X21 B9 B10 0 T3 T4 MM CELLSPS

X22 B10 B11 0 T3 T4 MM CELLSPS

```
X23  B11 B12  0 T3 T4 MM CELLSPS
X24  B12 B13   0 T3 T4 MM CELLSPS
*
*SECTION C CELLS
*
*
X25  C1 C2   0 T3 T4 MM CELLSPS
X26  C2 C3   0 T3 T4 MM CELLSPS
X27  C3 C4   0 T3 T4 MM CELLSPS
X28  C4 C5   0 T3 T4 MM CELLSPS
*
X29  C5 C6   0 T3 T4 MM CELLSPS
X30  C6 C7   0 T3 T4 MM CELLSPS
X31  C7 C8   0 T3 T4 MM CELLSPS
X32  C8 C9   0 T3 T4 MM CELLSPS
*
X33  C9 C10  0 T3 T4 MM CELLSPS
X34  C10 C11 0 T3 T4 MM CELLSPS
X35  C11 C12 0 T3 T4 MM CELLSPS
X36  C12 C13 0 T3 T4 MM CELLSPS
*
*SECTION D CELLS
*
*
X37  D1 D2   0 T3 T4 MM CELLSPS
X38  D2 D3   0 T3 T4 MM CELLSPS
X39  D3 D4   0 T3 T4 MM CELLSPS
X40  D4 D5   0 T3 T4 MM CELLSPS
*
X41  D5 D6   0 T3 T4 MM CELLSPS
X42  D6 D7   0 T3 T4 MM CELLSPS
X43  D7 D8   0 T3 T4 MM CELLSPS
X44  D8 D9   0 T3 T4 MM CELLSPS
*
X45  D9 D10  0 T3 T4 MM CELLSPS
X46  D10 D11 0 T3 T4 MM CELLSPS
X47  D11 D12 0 T3 T4 MM CELLSPS
X48  D12 D13 0 T3 T4 MM CELLSPS
X49  D13 D14 0 T3 T4 MM CELLSPS
X50  D14 D15 0 T3 T4 MM CELLSPS
X51  D15 D16 0 T3 T4 MM CELLSPS
X52  D16 D17 0 T3 T4 MM CELLSPS
X53  D17 D18 0 T3 T4 MM CELLSPS
X54  D18 D19 0 T3 T4 MM CELLSPS
X55  D19 D20 0 T3 T4 MM CELLSPS
X56  D20 D21 0 T3 T4 MM CELLSPS
X57  D21 D22 0 T3 T4 MM CELLSPS
X58  D22 D23 0 T3 T4 MM CELLSPS
X59  D23 D24 0 T3 T4 MM CELLSPS
X60  D24 D25 0 T3 T4 MM CELLSPS
X61  D25 D26 0 T3 T4 MM CELLSPS
X62  D26 D27 0 T3 T4 MM CELLSPS
X63  D27 D28 0 T3 T4 MM CELLSPS
X64  D28 D29 0 T3 T4 MM CELLSPS
X65  D29 D30 0 T3 T4 MM CELLSPS
X66  D30 D31 0 T3 T4 MM CELLSPS
X67  D31 D32 0 T3 T4 MM CELLSPS
X68  D32 D33 0 T3 T4 MM CELLSPS
X69  D33 D34 0 T3 T4 MM CELLSPS
X70  D34 D35 0 T3 T4 MM CELLSPS
```

```

X71  D35 D36  0 T3 T4 MM CELLSPS
X72  D36 D37  0 T3 T4 MM CELLSPS
X73  D37 D38  0 T3 T4 MM CELLSPS
X74  D38 D39  0 T3 T4 MM CELLSPS
X75  D39 D40  0 T3 T4 MM CELLSPS
X76  D40 D41  0 T3 T4 MM CELLSPS
X77  D41 D42  0 T3 T4 MM CELLSPS
X78  D42 D43  0 T3 T4 MM CELLSPS
X79  D43 D44  0 T3 T4 MM CELLSPS
X80  D44 D45  0 T3 T4 MM CELLSPS
X81  D45 D46  0 T3 T4 MM CELLSPS
X82  D46 D47  0 T3 T4 MM CELLSPS
X83  D47 D48  0 T3 T4 MM CELLSPS
X84  D48 D49  0 T3 T4 MM CELLSPS
X85  D49 D50  0 T3 T4 MM CELLSPS
X86  D50 D51  0 T3 T4 MM CELLSPS
X87  D51 D52  0 T3 T4 MM CELLSPS
X88  D52 D53  0 T3 T4 MM CELLSPS
X89  D53 D54  0 T3 T4 MM CELLSPS
X90  D54 D55  0 T3 T4 MM CELLSPS
X91  D55 D56  0 T3 T4 MM CELLSPS
X92  D56 D57  0 T3 T4 MM CELLSPS
X93  D57 D58  0 T3 T4 MM CELLSPS
X94  D58 D59  0 T3 T4 MM CELLSPS
X95  D59 D60  0 T3 T4 MM CELLSPS
X96  D60 D61  0 T3 T4 MM CELLSPS
X97  D61 D62  0 T3 T4 MM CELLSPS
X98  D62 D63  0 T3 T4 MM CELLSPS
X99  D63 D64  0 T3 T4 MM CELLSPS
X100 D64 D65  0 T3 T4 MM CELLSPS
X101 D65 D66  0 T3 T4 MM CELLSPS
X102 D66 D67  0 T3 T4 MM CELLSPS
X103 D67 D68  0 T3 T4 MM CELLSPS
X104 D68 D69  0 T3 T4 MM CELLSPS
X105 D69 D70  0 T3 T4 MM CELLSPS
X106 D70 D71  0 T3 T4 MM CELLSPS
X107 D71 D72  0 T3 T4 MM CELLSPS
X108 D72 D73  0 T3 T4 MM CELLSPS
*
*
* SUBCIRCUITS      SUBCIRCUITS
*
*
* Version of cell with current monitor for SPS Quad Cell
* These currents effect beam, to ground and damping resistor are not
* seen by beam
*
.SUBCKT CELLSPS T1 T2 T3 T4 T5 T6
* Bus between cells is not a good guess, just a placeholder
C1    T1 T3 14.7n
Rc1   T1 T3 86400k
VDFM  T1 E DC 0
X1    E T2 T3 SPSMAG
*
RD    T1 T2 24
* Ripple in Fermilab Booster
* FMEAS 0 T6 VDFM .0208
* Ripple in SPS
FMEAS 0 T6 VDFM 0.00925925925925926
RMEAS T6 0 1000000k

```

```

.ENDS CELLSPS
*
.SUBCKT SPSMAG 1 2 3
R1      1 4 14.2m
L1      4 2 13.9m
C2      2 3 14.7n
Rc2     2 3 86400k
.ENDS SPSMAG
*
.control
* AC analysis
* DEC (LIN)  Np  Fstart  Fstop
ac  dec    2000 .0001  1000k
* Transient analysis
*   TSTEP TSTOP <TSTART <TMAX>>
* tran 10u 10 8 200u
* set numdgt=16
set length=5000
set width=132
set nopage
* plot v(2) v(A4) v(A8) v(A12)
plot mag(v(3)/i(vmea) ) mag(v(3)/i(vbeam) )
plot ph(v(3)/i(vmea))*180/pi ph(v(3)/i(vbeam))*180/pi
print mag(v(3)/i(vmea)) ph(v(3)/i(vmea)) mag(v(3)/i(vbeam)) ph(v(3)/i(vbeam)) > SPS2_1000k.txt
* print mag(v(3)/i(vmea) ) ph(v(3)/i(vmea) ) mag(v(3)/i(vbeam) ) ph(v(3)/i(vbeam) ) > SPS2_1m.txt
.endc

```

Fig.B.3 shows that the models give the same results over a frequency range between $0.1\text{m}\Omega$ and $10\text{k}\Omega$.




Figure B.3: SPS Machine Impedance for Open and Closed SPS Circuit (Range 1m Hz - 10 kHz): Both calculated with Trans1 [2] & ngSPICE.




Figure B.4: SPS Impedance Calculation with ngSPICE for Closed SPS Circuit (Range 0.1m Hz - 1000 kHz): Both for all Machine and just Qd Quadrupoles.

Since the goal is to determine which current is actually flowing through the quadrupole chain only we have made the measurement at each of the 108 cells and add it all up. Fig. B.4 shows that at frequencies larger than some 40 Hz a significant part of the current either flows through the damping resistor or to ground. The impedance with respect to the beam is growing linearly in a double logarithmic scale. Therefore voltage ripple of larger than this 40 Hz is significantly reduced. As a result the ripple depth also diminishes at the same rate for equal voltage ripple amplitude.